**Zadání:** Vyjádřete analytický předpis funkce průhybové čáry, polohu a velikost maximálního průhybu prostě podepřeného nosníku excetricky zatíženého osamělou silou. Porovnejste s nosníkem zatíženým symetricky.




reakce v podporách

$$
\begin{aligned}
A &= \frac{1}{4}F \\
B &= \frac{3}{4}F
\end{aligned}
$$

```
eqv:A-F+B=0; /* vertical balance eq. */
eqM:A*2*L-F*(1/2)*L=0 ; /* moment bal. eq. */
linsolve([eqv,eqM],[A,B]); /* solve */
```

$$\overrightarrow{x_1} \in \left\langle 0, \frac{3}{2}L \right)$$

$$
\begin{aligned}
M(x_1) &= Ax_1 = \frac{1}{4}Fx_1 \\
EI_y\varphi(x_1) &= -\frac{1}{4}F\frac{x_1^2}{2} + C_1 = -\frac{1}{8}Fx_1^2 + C_1 \\
EI_y\mathrm{w}(x_1) &= -\frac{1}{8}F\frac{x_1^3}{3} + C_1x_1 + C_2 = -\frac{1}{24}Fx_1^3 + C_1x_1 + C_2
\end{aligned}
$$

$$\overleftarrow{x_2} \in \left\langle 0, \frac{1}{2}L \right)$$

$$
\begin{aligned}
M(x_2) &= Bx_2 = \frac{3}{4}Fx_2 \\
EI_y\varphi(x_2) &= -\frac{3}{4}F\frac{x_2^2}{2} + C_3 = -\frac{3}{8}Fx_2^2 + C_3 \\
EI_y\mathrm{w}(x_2) &= -\frac{3}{8}F\frac{x_2^3}{3} + C_3x_2 + C_4 = -\frac{1}{8}Fx_2^3 + C_3x_2 + C_4
\end{aligned}
$$

okrajové podmínky

$$
\begin{aligned}
\mathrm{w}_1(0) &= 0 = C_2 \\
\mathrm{w}_2(0) &= 0 = C_4 \\
\varphi_1\left(\frac{3}{2}L\right) &= -\varphi_2\left(\frac{1}{2}L\right) \\
-\frac{1}{8}F\left(\frac{3}{2}L\right)^2 + C_1 &= \frac{3}{8}F\left(\frac{1}{2}L\right)^2 - C_3 \\
\mathrm{w}_1\left(\frac{3}{2}L\right) &= \mathrm{w}_1\left(\frac{1}{2}L\right) \\
-\frac{1}{24}F\left(\frac{3}{2}L\right)^3 + C_1\frac{3}{2}L &= -\frac{1}{8}F\left(\frac{1}{2}L\right)^3 + C_3\frac{1}{2}
\end{aligned}
$$

řešením soustavy lineárních rovnic získáme

$$
\begin{aligned}
C_1 &= \frac{5}{32}FL^2 \\
C_3 &= \frac{7}{32}FL^2
\end{aligned}
$$

výsledek

$$
\begin{aligned}
EI_y \mathrm{w}(x_1) &= -\frac{1}{24}Fx_1^3 + \frac{5}{32}FL^2x_1 \\
EI_y \mathrm{w}(x_2) &= -\frac{1}{8}Fx_2^3 + \frac{7}{32}FL^2x_2
\end{aligned}
$$

polohu maximálního průhybu hledáme v intervalu $\overrightarrow{x_1} \in \left\langle 0, \frac{3}{2}L \right)$

$$
\begin{aligned}
EI_y\varphi(x_1) = 0 &= -\frac{1}{8}Fx_1^2 + \frac{5}{32}FL^2 \\
x_{\mathrm{w_{max}}} &= \frac{\sqrt{5}}{2}L = 1,118L
\end{aligned}
$$

maximální průhyb

$$
EI_y \mathrm{w_{max}} = EI_y \mathrm{w}\left(\frac{\sqrt{5}}{2}L\right) = -\frac{1}{24}\left(\frac{\sqrt{5}}{2}L\right)^3 + \frac{5}{32}FL^2\frac{\sqrt{5}}{2}L = \frac{5^{\frac{3}{2}}}{96}FL^3 = 0.116FL^3
$$






Řešení s využitím Maxima 5.32.1

```
/*
created 2020/10/24, kytyr@itam.cas.cz, Maxima 5.32.1
bending curve function
*/

kill(all); /* no mercy */
globalsolve: true;
eqv:A-F+B=0; /* vertical balance eq. */
eqM:A*2*L-F*(1/2)*L=0 ; /* moment balance eq. */
linsolve([eqv,eqM],[A,B]); /* solve linear system */

M1:A*x; /* M(x) in <0;(3/2)L) */
_phi1:-integrate(M1,x)+C1; /* angle */
_w1:integrate(_phi1,x)+C2; /* deflection */

M2:B*x; /* M(x) in <0;(1/2)L) */
_phi2:-integrate(M2,x)+C3; /* angle */
_w2:integrate(_phi2,x)+C4; /* deflection */

w10:subst(0,x,_w1); /* B.C. w(0) = 0 */
w20:subst(0,x,_w2); /* B.C. w(0) = 0 */
/* B.C. phi((3/2)L) = -phi(1/2)L) */
phi32L:subst((3/2)*L,x,_phi1);
phi12L:subst((1/2)*L,x,_phi2);
/* B.C. w((3/2)L) = w(1/2)L) */
w32L:subst((3/2)*L,x,_w1);
w12L:subst((1/2)*L,x,_w2);
linsolve([phi32L+phi12L,w32L-w12L,w10,w20],[C1,C2,C3,C4]);

phi1:-integrate(M1,x)+C1; /* phi(x) function in <0;(3/2)L) */
w1:integrate(phi1,x)+C2; /* w(x) function in <0;(3/2)L) */
phi2:-integrate(M2,x)+C3; /* phi(x) function in <0;(3/2)L) */
w2:integrate(phi2,x)+C4; /* w(x) function in <0;(3/2)L) */

loc:solve([phi1],[x]); /* position of max. deflection */
root1:rhs(loc[1]);
root2:rhs(loc[2]);
wmax:subst(root2,x,w1); /* max. deflection */
float(wmax);
```

Největší časovou úsporu a případnou elimanaci chyb v úpravě výrazů přináší použití SW nástrojů v tomto případě v místě řešení soustavy rovnic a vyjadřování maximálního průhybu.

Porovnání s nosníků zatížených exentricky a symetricky.




$$\overrightarrow{x_1} \in \left\langle 0, \frac{3}{2}L \right)$$

$$M(x_1) = \frac{1}{4}Fx_1$$

$$EI_y\mathrm{w}(x_1) = -\frac{1}{24}Fx_1^3 + \frac{5}{32}FL^2x_1$$

$$\overleftarrow{x_2} \in \left\langle 0, \frac{1}{2}L \right)$$

$$M(x_2) = \frac{3}{4}Fx_2$$

$$EI_y\mathrm{w}(x_2) = -\frac{1}{8}Fx_2^3 + \frac{7}{32}FL^2x_2$$




$$\overrightarrow{x_1} \in < 0, L) \equiv \overleftarrow{x_2} \in < 0, L)$$

$$M(x_{1,2}) = \frac{1}{2}Fx_{1,2}$$

$$EI_y\mathrm{w}(x_{1,2}) = -\frac{1}{6}qLx_{1,2}^3 + \frac{1}{2}qL^3x_{1,2}$$